# Maintainer: Thomas Dziedzic < gostrc at gmail >
# Contributor: Daniel J Griffiths <ghost1227@archlinux.us>
# Contributor: Evangelos Foutras <foutrelis@gmail.com>

pkgname=python-pyinotify
pkgver=0.9.6
pkgrel=12
pkgdesc='Python module used for monitoring filesystems events on Linux platforms with inotify.'
arch=('any')
url="https://github.com/seb-m/pyinotify"
license=('custom:MIT')
depends=('python')
makedepends=('python')
source=("https://github.com/seb-m/pyinotify/archive/${pkgver}.tar.gz")
sha512sums=('144db691c1cdfd4e52b838b8ed839b8d50d84d26a91a59b7460c3170f58a1c96b3ce4f51e6273835835291781215a87b8f2d2d87d1abf012e4c14b2cd4b3f4cf')

build() {
  cd pyinotify-${pkgver}
  python3 setup.py build
}

package() {
  cd pyinotify-${pkgver}
  python3 setup.py install --root="${pkgdir}" -O1

  install -d "${pkgdir}"/usr/share/licenses/python-pyinotify
  install -m644 COPYING \
    "${pkgdir}"/usr/share/licenses/python-pyinotify
}
